
1 Overview

This document lists all changes needed to correct problems in the TTCN implementation of test case 12.26 which is part of the IMS test suite in the ‘iwd-TTCN3-B2016-06_D16wk49’ ATS delivery.

The test case can be demonstrated to run on one IMS UE (see below). Execution log is provided as evidence.

2 Verification Test Summary

Test Case:
TC_12_26
ATS Version:
iwd-TTCN3-B2016-06_D16wk49
System Simulator used:
R&S CMW500
UE used:
Mediatek Whitney MT6799
Verification Status:
PASS

3 Corrections required for test case 12.26
Change 1 
	Function name
	f_TC_12_26_IMS1

	Reason for change
	1. The re-INVITE sent in step 14 is in-dialog and must be coded as such.
2. UE must attach an SDP body to 200 OK response sent in step 15. However, TTCN forbids such body when not listing a corresponding template in expected response.  
3. The re-INVITE procedure initiated in step 14 needs to be concluded by an ACK to be sent from SS to UE.

	Summary of change
	1. Added condition A5 in order to build proper SIP header.
2. Added cr_MessageBody_SDP to the expected response.  
3. Added sending of ACK. Note: the prose needs to be corrected for this as well. A draft prose CR is attached.

	TTCN module
	IMS/12/IMS_CC_CallControlTestcases.ttcn 

	MCC160 Comment
	


Before change

   function f_TC_12_26_IMS1() runs on IMS_PTC

  {

    var IMS_InviteRequestWithSdp_Type v_InviteRequestWithSdp;  /* @sic R5w140112: IMS_InviteRequestWithSdp_Type instead of IMS_DATA_REQ */

    var MessageHeader v_MessageHeader_Response183;

    var IMS_SDP_VoiceCodecList_Type v_AdditionalVoiceCodecList := {AMR_16000, EVS_16000};

    var template (value) MessageHeader v_MessageHeader_Invite;

    var template (present) MessageHeader v_MessageHeader_Response;

    var template (present) IMS_Response_Type v_ExpectedResponse;

    var SipUrl v_ContactUrl_UE;

    var SDP_Message v_SDP_LatestOffer;

    var template (value) SDP_Message v_SDP_Message;

    var template (value) INVITE_Request v_InviteRequest;

    var InternetProtocol_Type v_InternetProtocol;

    var IMS_DATA_RSP v_IMS_DATA_RSP;

    f_IMS_CC_Preamble(IPCAN_SpeechCall, IMS_REGISTERED);

    f_IMS_TestBody_Set(true);

    // @siclog "Step 1" siclog@

    f_IMS_CC_StartCall(IPCAN_MO_SpeechCall);

    // @siclog "Step 2" siclog@

    v_InviteRequestWithSdp := f_IMS_MOCallSetup_AnnexC21_Step2_A4(-, -, v_AdditionalVoiceCodecList);

    v_ContactUrl_UE := f_MessageHeader_GetContactSipUrl(v_InviteRequestWithSdp.Invite.msgHeader);

    // @siclog "Step 3 - 4" siclog@

    v_MessageHeader_Response183 := f_IMS_MOCallSetup_AnnexC21_Step3_4(v_InviteRequestWithSdp, -, -, -, EVS_16000);

    // See TS. 36.508 EUTRA/EPS signalling for IMS MO speech call Table 4.5A.6.3-1 Step 9-11

    f_IMS_CC_TriggerDedicatedBearerActivation(IPCAN_MO_SpeechCall);  // trigger IPCAN to continue with activation of secondary PDP context or dedicated EPS bearer

    // @siclog "Step 5 - 13" siclog@

    v_SDP_LatestOffer := f_IMS_MOCallSetup_AnnexC21_Steps5_13(v_InviteRequestWithSdp, v_MessageHeader_Response183, -, -, -, EVS_16000);

    // preparation of new SDP for Step 14

    v_SDP_Message := f_IMS_BuildSDP_AnnexC21C25_Step6or8(v_SDP_LatestOffer); // SDP offer from step 6 or 8

    v_SDP_Message := f_IMS_BuildSDP_CommonResponseToOfferTX(valueof(v_SDP_Message)); // new SDP offer for step 14: sess-version being incremented

    v_SDP_Message.media_list := fl_IMS_InjectSDP_FMTP(valueof(v_SDP_Message.media_list));

    // @siclog "Step 14" siclog@

    // Step 14 SS sends a re-INVITE with a second SDP offer to switch to EVS AMR-WB IO mode

    v_MessageHeader_Invite  := f_IMS_InviteRequest_MessageHeaderTX();        

    v_InviteRequest := cs_INVITE_Request(v_ContactUrl_UE, v_MessageHeader_Invite, cs_MessageBody_SDP(v_SDP_Message));

    v_InviteRequestWithSdp := f_IMS_InviteRequestWithSdpMT(v_InviteRequest, v_SDP_Message);

    IMS_Client.send(cas_IMS_Invite_Request(v_InviteRequestWithSdp.RoutingInfo, v_InviteRequest));

    // @siclog "Step 15" siclog@

    v_MessageHeader_Invite := v_InviteRequestWithSdp.Invite.msgHeader;

    v_MessageHeader_Response := f_IMS_OtherResponse_200_MessageHeaderRX(v_MessageHeader_Invite, cr_ContentTypeSDP);             // @sic R5-162926: Require header sic@

    v_InternetProtocol := v_InviteRequestWithSdp.RoutingInfo.Protocol;

    v_ExpectedResponse := cr_Response(c_statusLine200, v_MessageHeader_Response);

    v_IMS_DATA_RSP := f_IMS_ReceiveResponse_OptionalTrying(v_InviteRequestWithSdp.Invite, v_InternetProtocol, v_ExpectedResponse);   /* @sic R5-162046 sic@ @sic R5s160540: further improvement sic@ */

    f_IMS_SIP_SdpMessageBody_DecodeMatchAndCheckSDP(v_IMS_DATA_RSP.Response.messageBody, fl_IMS_BuildResponseSDP());

    // Step 16-17. SS sends BYE to release the call; UE sends 200 OK for the BYE request and ends the call

    // @siclog "Step 16 -17" siclog@

    f_IMS_CallReleaseMT(v_InviteRequestWithSdp.Invite, v_ContactUrl_UE);

    f_IMS_TestBody_Set(false);

    f_IMS_CC_Postamble(IPCAN_MO_SpeechCall);

  }
After change

  function f_TC_12_26_IMS1() runs on IMS_PTC

  {

    var IMS_InviteRequestWithSdp_Type v_InviteRequestWithSdp;  /* @sic R5w140112: IMS_InviteRequestWithSdp_Type instead of IMS_DATA_REQ */

    var MessageHeader v_MessageHeader_Response183;

    var IMS_SDP_VoiceCodecList_Type v_AdditionalVoiceCodecList := {AMR_16000, EVS_16000};

    var template (value) MessageHeader v_MessageHeader_Invite;

    var template (present) MessageHeader v_MessageHeader_Response;

    var template (present) IMS_Response_Type v_ExpectedResponse;

    var SipUrl v_ContactUrl_UE;

    var SDP_Message v_SDP_LatestOffer;

    var template (value) SDP_Message v_SDP_Message;

    var template (value) INVITE_Request v_InviteRequest;

    var InternetProtocol_Type v_InternetProtocol;

    var IMS_DATA_RSP v_IMS_DATA_RSP;
    var template (value) MessageHeader v_MessageHeader_Ack;
    f_IMS_CC_Preamble(IPCAN_SpeechCall, IMS_REGISTERED);

    f_IMS_TestBody_Set(true);

    // @siclog "Step 1" siclog@

    f_IMS_CC_StartCall(IPCAN_MO_SpeechCall);

    // @siclog "Step 2" siclog@

    v_InviteRequestWithSdp := f_IMS_MOCallSetup_AnnexC21_Step2_A4(-, -, v_AdditionalVoiceCodecList);

    v_ContactUrl_UE := f_MessageHeader_GetContactSipUrl(v_InviteRequestWithSdp.Invite.msgHeader);

    // @siclog "Step 3 - 4" siclog@

    v_MessageHeader_Response183 := f_IMS_MOCallSetup_AnnexC21_Step3_4(v_InviteRequestWithSdp, -, -, -, EVS_16000);

    // See TS. 36.508 EUTRA/EPS signalling for IMS MO speech call Table 4.5A.6.3-1 Step 9-11

    f_IMS_CC_TriggerDedicatedBearerActivation(IPCAN_MO_SpeechCall);  // trigger IPCAN to continue with activation of secondary PDP context or dedicated EPS bearer

    // @siclog "Step 5 - 13" siclog@

    v_SDP_LatestOffer := f_IMS_MOCallSetup_AnnexC21_Steps5_13(v_InviteRequestWithSdp, v_MessageHeader_Response183, -, -, -, EVS_16000);

    // preparation of new SDP for Step 14

    v_SDP_Message := f_IMS_BuildSDP_AnnexC21C25_Step6or8(v_SDP_LatestOffer); // SDP offer from step 6 or 8

    v_SDP_Message := f_IMS_BuildSDP_CommonResponseToOfferTX(valueof(v_SDP_Message)); // new SDP offer for step 14: sess-version being incremented

    v_SDP_Message.media_list := fl_IMS_InjectSDP_FMTP(valueof(v_SDP_Message.media_list));

    // @siclog "Step 14" siclog@

    // Step 14 SS sends a re-INVITE with a second SDP offer to switch to EVS AMR-WB IO mode

     v_MessageHeader_Invite  := f_IMS_InviteRequest_MessageHeaderTX(A_2_1_A5);

    v_InviteRequest := cs_INVITE_Request(v_ContactUrl_UE, v_MessageHeader_Invite, cs_MessageBody_SDP(v_SDP_Message));

    v_InviteRequestWithSdp := f_IMS_InviteRequestWithSdpMT(v_InviteRequest, v_SDP_Message);

    IMS_Client.send(cas_IMS_Invite_Request(v_InviteRequestWithSdp.RoutingInfo, v_InviteRequest));

    // @siclog "Step 15" siclog@

    v_MessageHeader_Invite := v_InviteRequestWithSdp.Invite.msgHeader;

    v_MessageHeader_Response := f_IMS_OtherResponse_200_MessageHeaderRX(v_MessageHeader_Invite, cr_ContentTypeSDP);             // @sic R5-162926: Require header sic@

    v_InternetProtocol := v_InviteRequestWithSdp.RoutingInfo.Protocol;

    v_ExpectedResponse := cr_Response(c_statusLine200, v_MessageHeader_Response, cr_MessageBody_SDP);

    v_IMS_DATA_RSP := f_IMS_ReceiveResponse_OptionalTrying(v_InviteRequestWithSdp.Invite, v_InternetProtocol, v_ExpectedResponse);   /* @sic R5-162046 sic@ @sic R5s160540: further improvement sic@ */

    f_IMS_SIP_SdpMessageBody_DecodeMatchAndCheckSDP(v_IMS_DATA_RSP.Response.messageBody, fl_IMS_BuildResponseSDP());

    v_MessageHeader_Ack := f_IMS_AckRequest_MessageHeaderTX(v_InviteRequestWithSdp.Invite);

    IMS_Client.send(cas_IMS_Ack_Request(cs_IMS_RoutingInfo(v_InviteRequestWithSdp.RoutingInfo.Protocol), cs_ACK_Request(v_ContactUrl_UE, v_MessageHeader_Ack)));
    // Step 16-17. SS sends BYE to release the call; UE sends 200 OK for the BYE request and ends the call

    // @siclog "Step 16 -17" siclog@

    f_IMS_CallReleaseMT(v_InviteRequestWithSdp.Invite, v_ContactUrl_UE);

    f_IMS_TestBody_Set(false);

    f_IMS_CC_Postamble(IPCAN_MO_SpeechCall);

  }
Change 2 
	Function name
	fl_IMS_InjectSDP_FMTP

	Reason for change
	Existing code replaces earlier fmtp line by “a=fmtp: evs-mode-switch=1” instead of just adding the evs mode switch indication to the fmtp line (which carries other entries like max-red, bw and mode-change-capability, and those should not be lost)

	Summary of change
	Take last fmtp line sent and add evs mode switch to it.

	TTCN module
	IMS/12/IMS_CC_CallControlTestcases.ttcn

	MCC160 Comment
	


Before change

   function fl_IMS_InjectSDP_FMTP(SDP_media_desc_list p_SDP_media_desc_list) return template (omit) SDP_media_desc_list

  {

    var template (value) SDP_attribute_rtpmap_codec v_RtpmapCodecInfo := cs_RTPMAP_EVS_16000;

    var template (value) SemicolonParam_List v_ParamList := { cs_EVS_ModeSwitch };
    var template (value) SDP_attribute_fmtp_format_specific_params v_Format_specific_params := cs_Fmtp_ParamList(v_ParamList);

    var template (present) SDP_attribute v_SDP_Attribute_anyfmtp := { fmtp := ? };

    var template (value) SDP_attribute v_SDP_Attribute_fmtp;

    var charstring v_Fmt;

    v_Fmt := f_SDP_MediaDescr_GetMediaFormat(p_SDP_media_desc_list[0], v_RtpmapCodecInfo);

    v_SDP_Attribute_fmtp := cs_SDP_Attribute_fmtp(v_Fmt, v_Format_specific_params);

    return f_IMS_SDP_MediaList_ExchangeAttribute(p_SDP_media_desc_list, v_SDP_Attribute_anyfmtp, v_SDP_Attribute_fmtp);

  }
After change

   function fl_IMS_InjectSDP_FMTP(SDP_media_desc_list p_SDP_media_desc_list) return template (omit) SDP_media_desc_list

  {

    var template (value) SDP_attribute_rtpmap_codec v_RtpmapCodecInfo := cs_RTPMAP_EVS_16000;

    var template (present) SDP_attribute_fmtp_format_specific_params v_FmtpParameters := f_SDP_FmtpDefaultParametersVoiceCodecRX(EVS_16000);

    var charstring v_FmtAudio := f_SDP_MediaDescr_GetVoiceCodec(p_SDP_media_desc_list[0], EVS_16000, v_FmtpParameters);

    var template (value) SemicolonParam_List v_FmtpParameterList_Step6_or_8 := f_SDP_MediaDescr_GetCodecParameters(p_SDP_media_desc_list[0].attributes, v_FmtAudio);
    var template (value)  SemicolonParam_List v_ParamList := f_SemicolonParam_List_Add_TX(v_FmtpParameterList_Step6_or_8,cs_EVS_ModeSwitch);   

    var template (value) SDP_attribute_fmtp_format_specific_params v_Format_specific_params := cs_Fmtp_ParamList(v_ParamList);

    var template (present) SDP_attribute v_SDP_Attribute_anyfmtp := { fmtp := ? };

    var template (value) SDP_attribute v_SDP_Attribute_fmtp;

    var charstring v_Fmt;

    v_Fmt := f_SDP_MediaDescr_GetMediaFormat(p_SDP_media_desc_list[0], v_RtpmapCodecInfo);

    v_SDP_Attribute_fmtp := cs_SDP_Attribute_fmtp(v_Fmt, v_Format_specific_params);

    return f_IMS_SDP_MediaList_ExchangeAttribute(p_SDP_media_desc_list, v_SDP_Attribute_anyfmtp, v_SDP_Attribute_fmtp);

  }
Change 3 
	Function name
	n/a   

	Reason for change
	Usage of f_SemicolonParam_List_Add_TX in Change 2 requires to import its definition

	Summary of change
	Import IMS_CommonTemplates

	TTCN module
	IMS/12/IMS_CC_CallControlTestcases.ttcn

	MCC160 Comment
	


Before change

    module IMS_CC_CallControlTestcases {

  import from LibSip_SIPTypesAndValues all;

  import from LibSip_SDPTypes all;

  import from LibSip_Common all;

  import from IP_ASP_TypeDefs all;

  import from IMS_Component all;

  import from IMS_ASP_TypeDefs all;

  import from IMS_ASP_Templates all;

  import from IMS_CommonDefs all;

  import from IMS_CommonParameters all;

  import from IMS_SIP_Templates all;

  import from IMS_MessageBody_Templates all;

  import from IMS_XML_Templates all;

  import from IMS_SDP_Templates all;

  import from IMS_SDP_Messages all;

  import from IMS_SDP_MessagesVideo all;

  import from IMS_CC_CommonFunctions all;

  import from IMS_Procedures_Registration all;

  import from IMS_Procedures_CallControl all;

  import from IMS_Procedures_Video all;

  import from IMS_Procedures_Common all;

  import from IMS_CommonFunctions all;

  import from IMS_PTC_CoordMsg all;

  import from CommonDefs all;

  import from UpperTesterFunctions  all;
…
After change

   module IMS_CC_CallControlTestcases {

  import from LibSip_SIPTypesAndValues all;

  import from LibSip_SDPTypes all;

  import from LibSip_Common all;

  import from IP_ASP_TypeDefs all;

  import from IMS_Component all;

  import from IMS_ASP_TypeDefs all;

  import from IMS_ASP_Templates all;

  import from IMS_CommonDefs all;

  import from IMS_CommonParameters all;

  import from IMS_SIP_Templates all;

  import from IMS_MessageBody_Templates all;

  import from IMS_XML_Templates all;

  import from IMS_SDP_Templates all;

  import from IMS_SDP_Messages all;

  import from IMS_SDP_MessagesVideo all;

  import from IMS_CC_CommonFunctions all;

  import from IMS_Procedures_Registration all;

  import from IMS_Procedures_CallControl all;

  import from IMS_Procedures_Video all;

  import from IMS_Procedures_Common all;

  import from IMS_CommonFunctions all;

  import from IMS_PTC_CoordMsg all;

  import from CommonDefs all;

  import from UpperTesterFunctions  all;

  import from IMS_CommonTemplates all;  
…
4 Execution Log Files
The Mediatek Whitney MT6799 UE passed this test case on R&S CMW500. The documentation below is enclosed as evidence of the successful test case run:

Test Case Execution log file: R5s170075
